ESKİŞEHİR OSMANGAZİ ÜNİVERSİTESİ
Mühendislik Mimarlık Fakültesi
İnşaat Mühendisliği Bölümü
E-Posta: ogu.ahmet.topcu@gmail.com
Web: http://mmf2.ogu.edu.tr/atopcu

Bilgisayar Destekli

Nümerik Analiz

Ders notları 2014

Ahmet TOPÇU



16

PROGRAMLAR: CholeskyBant2

16. PROGRAMLAR: Doğrusal denklem sistemi çözümü-CholeskyBant2[1]

Mühendislikte karşılaşılan problemlerin çoğunda $\underline{A}\ \underline{x}=\underline{b}$ denklem sisteminin $\underline{A}_{nxn}$ katsayılar matrisi simetrik ($\underline{A}^T=\underline{A}$), pozitif tanımlı ($\underline{x}^T\ \underline{A}\ \underline{x}>0$) ve bant şeklindedir.

$$\underline{A}_{nxn}\,\underline{x}_{nxm}=\underline{b}_{nxm}\rightarrow\begin{bmatrix} x & x & x & x & & & & & \\ & x & x & x & x & & & & \\ & & x & x & x & x & & & \\ & & & x & x & x & x & & \\ & & & & . & . & . & . & \\ & & & & & x & x & x & x \\ & & & & & & x & x & x \\ & & & & & & & x & x \\ & & & & & & & & x \end{bmatrix}\begin{bmatrix} x_{11} & x_{12} & \cdots & x_{1m} \\ x_{21} & x_{22} & \cdots & x_{2m} \\ & & & \\ & & & \\ . & . & . & . \\ & & & \\ & & & \\ & & & \\ x_{n1} & x_{n2} & & x_{nm} \end{bmatrix}=\begin{bmatrix} b_{11} & b_{12} & \cdots & b_{1m} \\ b_{21} & b_{22} & \cdots & b_{2m} \\ & & & \\ & & & \\ . & . & . & . \\ & & & \\ & & & \\ & & & \\ b_{n1} & b_{n2} & & b_{nm} \end{bmatrix}$$

Bilgisayar belleğinden yer kazanmak amacıyla $\underline{A}$ nın sadece bant içindeki sayıları n · iBant boyutlu dikdörtgen matrise depolanır:



Yukarıdaki şekilde *x* bant içinde ve sıfırdan farklı olduğu varsayılan terimleri göstermektedir. iBant bant genişliğidir, diyagonal ve diyagonali izleyen sıfırdan farklı terim sayısıdır: 1≤iBant≤n dir.

Bu tür denklem sisteminin çözümü için CholeskyBant2 alt programı kullanılabilir. Ekte QBASIC kodu verilen programda $\underline{A}_{nxn}$ simetrik matrisinin bant içindeki terimleri $\underline{A}_{nxiBant}$ matrisinde, $\underline{b}_{nxm}$ nin terimleri nxm matriste depolanmış olmalıdır.

CholeskyBant2 alt programı $\underline{x}$ bilinmeyenler matrisini ve ayrıca d, k gibi iki değer hesaplar. $\underline{A}$ nın determinantı

Det $\underline{A}=d\cdot 2^k$

ile hesaplanabilir. d=0 olması durumunda $\underline{A}$ tekildir veya pozitif tanımlı değildir, çözüm yoktur.

Hesaplanan $\underline{x}$ çözümü b matrisinde depolanır. Bu nedenle $\underline{x}$ matrisi için programda boyut açılmamıştır. Hesap sonrası $\underline{A}$ ve $\underline{b}$ nin ilk değerleri kaybolur. $\underline{A}=\underline{U}^T\underline{U}$ çarpanlarından $\underline{U}$ üst üçgeni(bant formunda) A matrisinde depolanır.

$\underline{b}=\underline{I}$ birim matris ve m=n olarak verilirse çözüm $\underline{A}^{-1}$ olur.

[1] Teori ve sayısal örnekler için bak: bölüm 6

Örnek:

$$
\underline{A}\underline{x}=\underline{b}\rightarrow
\overset{\longleftarrow\ \text{iBant=4}\ \longrightarrow}{\begin{bmatrix}
10 & 2 & -3 & -1 & & & \\
 & 9 & 2 & -1 & 3 & & \\
 & & 12 & 3 & 0 & 5 & \\
 & & & 8 & 1 & -2 & -2 \\
 & \text{Simetrik} & & & 6 & 1 & 2 \\
 & & & & & 9 & 3 \\
 & & & & & & 10
\end{bmatrix}}
\begin{bmatrix}
x_{11} & x_{12} \\
x_{21} & x_{22} \\
x_{31} & x_{32} \\
x_{41} & x_{42} \\
x_{51} & x_{52} \\
x_{61} & x_{62} \\
x_{71} & x_{72}
\end{bmatrix}
=
\begin{bmatrix}
10 & 10 \\
1 & -10 \\
7 & 7 \\
-11 & -11 \\
3 & 3 \\
8 & 8 \\
14 & 14
\end{bmatrix}
\rightarrow \underline{x}=?
$$

Katsayılar matrisi A nın programda depolanışı→
$$
\overset{\longleftarrow\ \text{iBant=4}\ \longrightarrow}{\begin{bmatrix}
10 & 2 & -3 & -1 \\
9 & 2 & -1 & 3 \\
12 & 3 & 0 & 5 \\
8 & 1 & -2 & -2 \\
6 & 1 & 2 & 0 \\
9 & 3 & 0 & 0 \\
10 & 0 & 0 & 0
\end{bmatrix}}
$$

Son satırlara eklenen sıfırlar

CholeskyBant2 sonucu

```
C:\NANALIZ\Basic\QBasic.EXE
Determinant ve kondisyon hesabı için katsayılar:
d= .815167427062988          k= 20

Denklem sisteminin çözümü (CholeskyBant2):
 2             -2              3             -3              2
-2              1

 2.78314507496213             -4.43241475727247             4.42785794926091
-4.31695261270642              3.71759138476657            -3.19383456271607
 .751241569320222
```

Çözüm:

$$
\underline{x}=\begin{bmatrix}
2 & 2.78 \\
-2 & -4.43 \\
3 & 4.43 \\
-3 & -4.32 \\
2 & 3.72 \\
-2 & -3.19 \\
1 & 0.75
\end{bmatrix}
$$

Ana program

```
'--------- Ana program CholeskyBant2-----------
' Ax=b denklem sisteminin çözümü
' A simetrik pozitif tanınlı bant matris
' Çağrılan programlar: CholeskyBant
'-------------------------------------------
DATA 7: 'Denklem sayısı n
DATA 4: 'yarı bant genişliği iBant
DATA 2: 'karşı taraf sayısı m
'A matrisi:
DATA 10,2,-3,-1
DATA 9,2,-1,3
DATA 12,3,0,5
DATA 8,1,-2,-2
DATA 6,1,2,0
DATA 9,3,0,0
DATA 10,0,0,0
'b matrisi:
DATA 10,1,7,-11,3,8,14
DATA 10,-10,7,-11,3,8,14

DEFINT I-N
DEFDBL A-H, O-Z

DECLARE SUB CholeskyBant2 (n, iBant, a(), m, b(), d, k)
    CLS
    READ n, iBant, m
    DIM a(n, iBant), b(n, m)

'A yı oku:
    FOR i = 1 TO n
     FOR j = 1 TO iBant
      READ a(i, j)
     NEXT j
    NEXT i

'b yi oku:
    FOR i = 1 TO m
     FOR j = 1 TO n
      READ b(j, i)
     NEXT j
    NEXT i

    CALL CholeskyBant2(n, iBant, a(), m, b(), d, k)

    PRINT "Determinant ve kondisyon hesabı için katsayılar:"
    PRINT "d="; d, "k="; k
    IF d = 0 THEN PRINT "Matris pozitif tanımlı değil !": END
    PRINT
    PRINT "Denklem sisteminin çözümü (CholeskyBant2):"
    FOR i = 1 TO m
     FOR j = 1 TO n
      PRINT b(j, i),
     NEXT j
     PRINT : PRINT
    NEXT i

END ' CholeskyBant ana
```

Denklem sayısı, bant genişliği ve karşı taraf sayısı verileri

A nın üst üçgeninin bant içindeki satırları

B nin kolonları

Denklem sayısı, bant genişliği ve karşı taraf sayısı okunuyor

A ve B matrisleri için bellekte yer ayrılıyor

A nın bant içindeki satırları DATA satırlarından okunuyor

B nin kolonları DATA satırlarından okunuyor

CholeskyBant2 alt programı çağrılıyor

Sonuçlar yazdırılıyor

CholeskyBant2 alt programı

```
SUB CholeskyBant2 (n, iBant, a(), m, b(), d, k)
'--------------------------------------------------------
' Doğrusal denklem sistemi çözümü
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994
' A(n,n)*x(n,m)=b(n,m) sisteminin x(n,m) vektorü hesaplanır.
' n denklem sayısı, iBant bant genişliğidir.
' A katsayılar matrisi simetrik ve pozitif tanımlı olmalıdır.
' A nın diyagonal ve sağındaki, bant içindeki, elemanları
' A(n,iBant) alanında depolanmış olmalıdır.
' b matrisi depolanmış olmalıdır.
' x(n,m) çözümü b(n,m) matrisinde depolanır.
' A nın determinantı det a=d*2^k ile hesaplanabilir.
' Program d=0 değeri ile dönerse, a matrisi tekildir veya
' pozitif tanımlı değildir ve CholeskyBant2 çözüm vermez.
'--------------------------------------------------------
' Machep
  Eps = 1
  DO
   Eps = Eps / 2
   s = 1 + Eps
  LOOP UNTIL s <= 1
  Eps = 2 * Eps
' Zero: sıfır varsayılacak sayı
    Zero = 0
    FOR i = 1 TO n
     IF ABS(a(i, 1)) >= Zero THEN Zero = ABS(a(i, 1))
    NEXT i
    Zero = Zero * Eps

    iBant1 = iBant - 1
' Çarpanlara ayır
    FOR i = 1 TO n
     IF a(i, 1) <= Zero GOTO 100
     T = SQR(a(i, 1))
     FOR j = 1 TO iBant
      a(i, j) = a(i, j) / T
     NEXT j
     FOR j = 1 TO iBant1
     i1 = i + j
      IF i1 > n THEN GOTO 200
      T = a(i, j + 1)
      FOR k = j TO iBant1
       iv = 1 + k - j
       a(i1, iv) = a(i1, iv) - T * a(i, k + 1)
      NEXT k
     NEXT j
200  NEXT i

' Yukarıdan aşağı hesap
    FOR m1 = 1 TO m
     FOR i = 1 TO n
      b(i, m1) = b(i, m1) / a(i, 1)
      FOR j = 1 TO iBant1
       i1 = i + j
       IF i1 > n GOTO 300
       T = a(i, j + 1)
       b(i1, m1) = b(i1, m1) - T * b(i, m1)
      NEXT j
300   NEXT i
    NEXT m1

' Aşağıdan yukarı hesap
    FOR m1 = 1 TO m
     b(n, m1) = b(n, m1) / a(n, 1)
     FOR i = 1 TO n - 1
      k = n - i
      T = 0
      FOR j = 1 TO iBant1
       i1 = j + k
       IF n < i1 GOTO 400
       T = T + a(k, j + 1) * b(i1, m1)
      NEXT j
400    b(k, m1) = (b(k, m1) - T) / a(k, 1)
     NEXT i
    NEXT m1

' determinant (WILKINSON)
    d = 1: k = 0
    FOR i = 1 TO n
     d = d * a(i, 1) * a(i, 1)
500   k = k + 4
     IF ABS(d) >= 1 THEN d = d * .0625: GOTO 500
600   k = k - 4
     IF ABS(d) < .0625 THEN d = d * 16: GOTO 600
    NEXT i
    EXIT SUB

100 'Matris tekil veya pozitif tanımlı değil
    d = 0: k = 0

END SUB 'CholeskyBant2
```